// File: contracts/OTNFTBonusIssuer.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import "./OTCallable.sol";

interface IOTMint {
    function mint(address to, uint256 amount) external;
}

contract OTNFTBonusIssuer is OTCallable {
    mapping(address => uint256) public bonusAllocation;
    IOTMint public openTownContract;

    event BonusAllocationChanged(
        address indexed by,
        address indexed nftContract,
        uint256 amount
    );
    event BonusIssued(address indexed by, address indexed to, uint256 amount);

    constructor(address otContractAddr) {
        require(otContractAddr != address(0), "Invalid address");
        openTownContract = IOTMint(otContractAddr);
    }

    function setBonusAllocation(address addr, uint256 amount)
        external
        onlyOwner
    {
        bonusAllocation[addr] = amount;
        emit BonusAllocationChanged(msg.sender, addr, amount);
    }

    function mintBonus(address to, uint256 numTokens) external onlyOTCaller {
        uint256 bonusAmount = bonusAllocation[msg.sender] * numTokens;
        require(bonusAmount > 0, "Invalid bonus amount");

        emit BonusIssued(msg.sender, to, bonusAmount);
        openTownContract.mint(to, bonusAmount);
    }
}

// File: contracts/OTCallable.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import "@openzeppelin/contracts/access/Ownable.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is a set of accounts (authorized callers) that can be granted exclusive
 * access to specific functions.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOTCaller`, which can be applied to your functions to restrict their use
 * to authorized Open Town accounts.
 */
abstract contract OTCallable is Ownable {
    mapping(address => bool) private otCallers;

    event CallerAuthorized(address indexed caller, address indexed by);
    event CallerDeauthorized(address indexed caller, address indexed by);

    /**
     * @dev Throws if called by unauthorized account.
     */
    modifier onlyOTCaller() {
        require(otCallers[msg.sender], "OTCallable: Unauthorized caller");
        _;
    }

    /**
     * @dev Authorizes new account (`addr`).
     * Can only be called by the current owner.
     */
    function authorizeCaller(address addr) external onlyOwner {
        otCallers[addr] = true;
        emit CallerAuthorized(addr, msg.sender);
    }

    /**
     * @dev Revokes authorization from the account (`addr`).
     * Can only be called by the current owner.
     */
    function deauthorizeCaller(address addr) external onlyOwner {
        otCallers[addr] = false;
        emit CallerDeauthorized(addr, msg.sender);
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}